

Analysis of Wikidata Queries

Daniel Hernandez

Claudio Gutierrez

Aidan Hogan

April 6

In this document we analyse current Wikidata's example queries that have the possibility of include false positives in terms of certain answers when blank nodes are interpreted with the semantics of existential variables.

1 Queries

We classify queries in four classes. Class A include all queries such that we can build a database where the query has false positives. Class B include all queries where we can build a database having only blank nodes in object position of predicates. This assumption is motivated by the nature of Wikidata, where all blank nodes occur in such position. Class C include all queries where the involved predicated have blank nodes in an object position. Finally, class D include all queries that have false positives in the current instance of Wikidata.

In this section we only analyse queries in class A (note that classes B, C and D are subclasses of class A).

1.1 Women with most sitelinks and no image born in 1921 or later

```
SELECT ?s ?desc ?linkcount
WHERE
{
  ?s wdt:P31 wd:Q5 ;
    wdt:P21 wd:Q6581072 ;
    wdt:P569 ?born .
  FILTER (?born >= "1921-01-01T00:00:00Z"^^xsd:dateTime) .
  ?s wikibase:sitelinks ?linkcount .
  MINUS { ?s wdt:P18 [] }
  ?s rdfs:label ?desc FILTER(lang(?desc)="en") .
}
GROUP BY ?s ?desc ?linkcount
ORDER BY DESC(?linkcount)
LIMIT 50
```

A. Differences in any database: Yes, because MINUS.

B. Differences assuming BNs only as object: Not, because there is no constant in MINUS nor correlated variable in object position.

C. Differences knowing what properties has BNs in Wikidata: Not.

D. Differences in Wikidata: Not.

1.2 Contemporary Indian actresses

```

SELECT ?item ?itemLabel ?itemDescription (SAMPLE(?img) AS ?image) (SAMPLE(?dob) AS ?dob) ?s1
WHERE {
    ?item wdt:P106 wd:Q33999 ;
        wdt:P27 wd:Q668 ;
        wdt:P21 wd:Q6581072 .
    MINUS { ?item wdt:P570 [] }
    OPTIONAL { ?item wdt:P18 ?img }
    OPTIONAL { ?item wdt:P569 ?dob }
    OPTIONAL { ?item wikibase:sitelinks ?s1 }
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en"}
} GROUP BY ?item ?itemLabel ?itemDescription ?s1
ORDER BY DESC(?s1)

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Not, because there is no constant in MINUS nor correlated variable in object position.
- C. Differences knowing what properties has BNs in Wikidata:** Not.
- D. Differences in Wikidata:** Not.

1.3 Map of places mentioned in travel stories with text in French accessible online

```

SELECT DISTINCT ?item ?itemLabel ?ed ?edLabel ?auteurLabel
                (year(?dateCreation) as ?AnneeCreation) ?lieuLabel ?coord ?lien
WHERE
{
    ?item wdt:P136 wd:Q1164267 .
    ?item wdt:P31 wd:Q571 .
    OPTIONAL{
        ?item wdt:P50 ?auteur .
    }
    OPTIONAL{
        ?item wdt:P571 ?dateCreation .
    }

    ?item wdt:P840 ?lieu .
    MINUS {?item wdt:P840 wd:Q933}
    ?lieu wdt:P625 ?coord .

    {
        ?item wdt:P953 ?lien .
        ?item wdt:P407 wd:Q150
    }UNION{
        ?ed wdt:P629 ?item .
        ?ed wdt:P953 ?lien .
        ?ed wdt:P407 wd:Q150
    }

    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],fr". }
} ORDER BY ?dateCreation

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Yes, because constant `wd:Q933` inside the FILTER may match a blank node.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because predicate `wdt:P840` has blank nodes in object position.
- D. Differences in Wikidata:** Not, because the following query returns false.

```
ASK {
  ?item wdt:P136 wd:Q1164267 .
  ?item wdt:P31 wd:Q571 .
  ?item wdt:P840 ?lieu .
  ?item wdt:P840 ?x .
  FILTER(isBlank(?x))
}
```

1.4 Streets in France without a city

```
SELECT ?street ?streetLabel
WHERE
{
  ?street wdt:P31/wdt:P279* wd:Q79007 .
  ?street wdt:P17 wd:Q142 .
  MINUS { ?street wdt:P131 [] } .
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" }
}
ORDER BY ?streetLabel
```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Not, because variable `?street` occurring in the MINUS also occurs as subject in the positive pattern outside the MINUS.
- C. Differences knowing what properties has BNs in Wikidata:** Not.
- D. Differences in Wikidata:** Not

1.5 ATMs around Munich belonging to the Bankcard-Servicenetz interbank network

```
PREFIX lgdo: <http://linkedgedata.org/ontology/>
PREFIX geom: <http://geovocab.org/geometry#>
PREFIX bif: <bif:>

SELECT ?atm ?geometry ?bank ?bankLabel WHERE {
  hint:Query hint:optimizer "None".

  SERVICE <http://linkedgedata.org/sparql> {
    { ?atm a lgdo:Bank; lgdo:atm true. }
```

```

UNION { ?atm a lgdo:Atm. }

?atm geom:geometry [geo:asWKT ?geometry];
      lgdo:operator ?operator.
FILTER(bif:st_intersects(?geometry, bif:st_point(11.5746898, 48.1479876), 5))
}

BIND(STRLANG(?operator, "de") as ?bankLabel)
?bank rdfs:label ?bankLabel.

{ ?bank wdt:P527 wd:Q806724. }
UNION { ?bank wdt:P1454 wd:Q5349747. }
MINUS { wd:Q806724 wdt:P3113 ?bank. }
}

```

- A. Differences in any database: Yes, because MINUS.
- B. Differences assuming BNs only as object: Not, because variable ?bank occurring in the MINUS also occurs as subject in the positive pattern outside the MINUS.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not

1.6 Objects with most mass

```

SELECT ?object ?objectLabel (MAX(?mass) AS ?mass)
WHERE
{
  ?object p:P2067/psv:P2067 [
    wikibase:quantityAmount ?baseMass;
    wikibase:quantityUnit [
      p:P2370/psv:P2370 [
        wikibase:quantityAmount ?unitMass;
        wikibase:quantityUnit wd:Q11570
      ]
    ]
  ]
  BIND(?baseMass * ?unitMass AS ?mass).
  MINUS { ?object wdt:P31 wd:Q3647172. }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
}
GROUP BY ?object ?objectLabel
ORDER BY DESC(?mass)

```

- A. Differences in any database: Yes, because MINUS.
- B. Differences assuming BNs only as object: Yes, constant wd:Q3647172 in object position in MINUS clause may match a blank.
- C. Differences knowing what properties has BNs in Wikidata: Yes, wdt:P31 has blank nodes in the object position.
- D. Differences in Wikidata: No, because the answer of the following query is false.

```

ASK {
  ?object p:P2067/psv:P2067 [
    wikibase:quantityAmount ?baseMass;
    wikibase:quantityUnit [
      p:P2370/psv:P2370 [
        wikibase:quantityAmount ?unitMass;
        wikibase:quantityUnit wd:Q11570
      ]
    ]
  ]
  ].
BIND(?baseMass * ?unitMass AS ?mass).
MINUS { ?object wdt:P31 ?x. FILTER (isBlank(?x)) }
SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
}

```

1.7 Periods of Japanese history and what they were named after

```

SELECT ?era ?eraLabel (YEAR(?start_time) AS ?start) (YEAR(?end_time) AS ?end)
       ?namedLabel ?namedDescription
WHERE {
  ?era wdt:P31 wd:Q6428674; wdt:P361 wd:Q130436;
       wdt:P580 ?start_time.
  MINUS { ?era wdt:P2348/wdt:P361 wd:Q130436 }
  OPTIONAL { ?era wdt:P582 ?end_time }
  OPTIONAL { ?era wdt:P138 ?named }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en" }
} ORDER BY ?start DESC(?end)

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Yes, constant `wd:Q130436` in object position in MINUS clause may match a blank, and it could be a blank in object position of predicate `wdt:P2348` in MINUS.
- C. Differences knowing what properties has BNs in Wikidata:** `wdt:P2348` does not have blank nodes in the object position, but `wdt:P361` has. Hence, the answer is yes.
- D. Differences in Wikidata:** No, because the answer of the following query is false.

```

ASK {
  ?era wdt:P31 wd:Q6428674; wdt:P361 wd:Q130436;
       wdt:P580 ?start_time.
  { ?era wdt:P2348/wdt:P361 ?x . FILTER (isBlank(?x)) }
}

```

1.8 Actors who played the same role more than 40 years apart

```

SELECT DISTINCT ?actor ?actorLabel ?characterLabel ?movie1Label ?movie2Label
WHERE {
  ?movie1 p:P161 [
    ps:P161 ?actor;
    pq:P453 ?character
  ];
  wdt:P577 ?movie1Publication.
  ?movie2 p:P161 [
    ps:P161 ?actor;
    pq:P453 ?character
  ];
  wdt:P577 ?movie2Publication.
  MINUS{?movie1 wdt:P31/wdt:P279? wd:Q24856}
  MINUS{?movie2 wdt:P31/wdt:P279? wd:Q24856}
  FILTER(?character != wd:Q18086706).
  FILTER(?movie1Publication + "P40Y"^^xsd:duration < ?movie2Publication)
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
}

```

A. Differences in any database: Yes, because MINUS.

B. Differences assuming BNs only as object: Yes, because one of the following cases can occur:

1. Variable `?character` occurs in object position of predicate `pq:P453` and in MINUS.
2. Constant `wd:Q24856` in object position in MINUS clause may match a blank.
3. It could be a blank inside the property path `wdt:P31/wdt:P279?`.
4. A blank is compared with constant `wd:Q18086706` in the inequity.
5. Variables `?movie1Publication` and `?movie2Publication` occur in inequality and are objects of predicate `wdt:P577`.

C. Differences knowing what properties has BNs in Wikidata: We analyze each case: Regarding case 1: yes, because `pq:P453` has blanks in object position. Regarding cases 2 and 3: yes, because `wdt:P31` and `wdt:P279?` have BNs in object position. Regarding case 4: yes, because predicate `wdt:P577` has blanks in object position. Hence, answer is yes.

D. Differences in Wikidata: Not because the result of the following query is false.

```

ASK {
  ?movie1 p:P161 [
    ps:P161 ?actor;
    pq:P453 ?character
  ];
  wdt:P577 ?movie1Publication.
  ?movie2 p:P161 [
    ps:P161 ?actor;
    pq:P453 ?character
  ];
  wdt:P577 ?movie2Publication.
  FILTER(isBlank(?movie1Publication) || isBlank(?movie2Publication))

  MINUS{?movie1 wdt:P31/wdt:P279? wd:Q24856}
  MINUS{?movie2 wdt:P31/wdt:P279? wd:Q24856}
}

```

```

FILTER(?character != wd:Q18086706).
FILTER(?movie1Publication + "P40Y"^^xsd:duration < ?movie2Publication)
}

```

```

SELECT DISTINCT ?actor ?actorLabel ?characterLabel ?movie1Label ?movie2Label
WHERE {
  ?movie1 p:P161 [
    ps:P161 ?actor;
    pq:P453 ?character
  ];
  wdt:P577 ?movie1Publication.
  ?movie2 p:P161 [
    ps:P161 ?actor;
    pq:P453 ?character
  ];
  wdt:P577 ?movie2Publication.
  FILTER(!isBlank(?character))
  FILTER(!isBlank(?movie1Publication))
  FILTER(!isBlank(?movie2Publication))

  MINUS{?movie1 wdt:P31/wdt:P279? ?b1 . FILTER(isBlank(?b1)) }
  MINUS{?movie2 wdt:P31/wdt:P279? ?b2 . FILTER(isBlank(?b2)) }
  MINUS{?movie1 wdt:P31/wdt:P279? wd:Q24856}
  MINUS{?movie2 wdt:P31/wdt:P279? wd:Q24856}
  FILTER(?character != wd:Q18086706).
  FILTER(?movie1Publication + "P40Y"^^xsd:duration < ?movie2Publication)
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
}

```

1.9 Sandwich ingredients

```

SELECT ?sandwich ?ingredient ?sandwichLabel ?ingredientLabel
WHERE
{
  ?sandwich wdt:P31?/wdt:P279* wd:Q28803;
    wdt:P527 ?ingredient.
  MINUS { ?ingredient wdt:P279* wd:Q7802. }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en", "fr". }
}
ORDER BY UCASE(STR(?sandwichLabel))

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Yes, because constant in object position inside MINUS, and because ?ingredient occurs in object position.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, wdt:P527 has BNs in object position.
- D. Differences in Wikidata:** No, because queries below return false.

```

ASK {
  ?sandwich wdt:P31?/wdt:P279* wd:Q28803;
            wdt:P527 ?ingredient.
  FILTER (isBlank(?ingredient))
}

```

```

ASK {
  ?sandwich wdt:P31?/wdt:P279* wd:Q28803;
            wdt:P527 ?ingredient.
  ?ingredient wdt:P279* ?x .
  FILTER (isBlank(?x))
}

```

1.10 Odd countries

```

SELECT ?item ?country ?itemLabel ?countryLabel
WHERE
{
  ?item wdt:P17 ?country.
  MINUS { ?country wdt:P31/wdt:P279* wd:Q7275. }
  MINUS { ?country wdt:P31/wdt:P279* wd:Q183366. }
  MINUS { ?country wdt:P31/wdt:P279* wd:Q15893266. }
  MINUS { ?country wdt:P31/wdt:P279* wd:Q1048835. }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
LIMIT 1000

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Yes, because constant in object position inside MINUS, and because ?country occurs in object position.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because wdt:P17 has blanks in object position, so ?country can be a blank node, and wdt:P279 has BNs in object position.
- D. Differences in Wikidata:** Yes, because the possible blank of ?country.

1.11 Capitals that aren't capitals

```

SELECT ?capital ?capitalLabel ?ofLabel ?start ?end
WHERE
{
  ?capital p:P31 ?capitalStat.
  ?capitalStat ps:P31 wd:Q5119.
  OPTIONAL { ?capitalStat pq:P642 ?of. }
  OPTIONAL { ?capitalStat pq:P580 ?start. }
}

```



```

OPTIONAL { ?capitalStat pq:P582 ?end. }
MINUS { [] wdt:P36 ?capital. }
SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Yes, because correlated variable `?capital` is in object position inside MINUS.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because `wdt:P36` has blanks in object position.
- D. Differences in Wikidata:** Yes, because `wdt:P36` has blanks in object position.

1.12 Look up q-codes based on English description

```

SELECT ?lab ?item WHERE {
  ?item rdfs:label ?lab .
  MINUS {?item wdt:P31 wd:Q4167836 } .
  VALUES ?lab {
    "Agelenopsis pennsylvanica"@en
    "Agelenopsis potteri"@en
    "Anelosimus studiosus"@en
  } .
}

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Yes, because constant in object position inside MINUS.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because `wdt:P31` has blanks in object position.
- D. Differences in Wikidata:** No, because the following query returns false.

```

ASK {
  ?item rdfs:label ?lab .
  ?item wdt:P31 ?x . FILTER(isBlank(?x)) .
  VALUES ?lab {
    "Agelenopsis pennsylvanica"@en
    "Agelenopsis potteri"@en
    "Anelosimus studiosus"@en
  }
}

```

1.13 Metabolites and the species where they are found in

```

PREFIX pr: <http://www.wikidata.org/prop/reference/>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX prov: <http://www.w3.org/ns/prov#>

SELECT ?compound ?compoundLabel ?speciesLabel ?sourceLabel ?doi ?wpid WHERE {
  ?compound wdt:P31 wd:Q11173.
  MINUS { ?compound wdt:P31 wd:Q8054. }
  ?compound p:P703 ?statement.
  ?statement rdf:type wikibase:BestRank.
  ?statement ps:P703 ?species.
  OPTIONAL {
    ?statement (prov:wasDerivedFrom/pr:P248) ?source.
    OPTIONAL { ?statement (prov:wasDerivedFrom/pr:P2410) ?wpid. }
    OPTIONAL { ?statement (prov:wasDerivedFrom/pr:P356) ?doi. }
  }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
ORDER BY ASC(?compound)

```

- A. Differences in any database:** Yes, because MINUS.
- B. Differences assuming BNs only as object:** Yes, because constant in object position inside MINUS.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because wdt:P31 has blanks in object position.
- D. Differences in Wikidata:** No, because the following query returns false.

```

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
ASK {
  ?compound wdt:P31 wd:Q11173.
  ?compound wdt:P31 ?x. FILTER (isBlank(?x))
  ?compound p:P703 ?statement.
  ?statement rdf:type wikibase:BestRank.
  ?statement ps:P703 ?species.
}

```

1.14 Largest cities with female mayor

```

SELECT DISTINCT ?city ?cityLabel ?mayor ?mayorLabel
WHERE
{
  BIND(wd:Q6581072 AS ?sex)
  BIND(wd:Q515 AS ?c)

  ?city wdt:P31/wdt:P279* ?c .
  ?city p:P6 ?statement .
  ?statement ps:P6 ?mayor .
  ?mayor wdt:P21 ?sex .
}

```

```

    FILTER NOT EXISTS { ?statement pq:P582 ?x }

    ?city wdt:P1082 ?population .
    SERVICE wikibase:label {
        bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en" .
    }
}
ORDER BY DESC(?population)
LIMIT 10

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object: Not, because correlated variable ?statement occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.15 List of countries ordered by the number of their cities with female mayor

```

SELECT ?country ?countryLabel (count(*) AS ?count)
WHERE
{
    ?city wdt:P31/wdt:P279* wd:Q515 .
    ?city p:P6 ?statement .
    ?statement ps:P6 ?mayor .
    ?mayor wdt:P21 wd:Q6581072 .
    FILTER NOT EXISTS { ?statement pq:P582 ?x }
    ?city wdt:P17 ?country .

    SERVICE wikibase:label {
        bd:serviceParam wikibase:language "ru,en" .
    }
}
GROUP BY ?country ?countryLabel
ORDER BY DESC(?count)
LIMIT 100

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object: Not, because correlated variable ?statement occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.16 Wikidata people per million inhabitants for all EU countries

```

SELECT
    ?country ?countryLabel
    ?wikiPersons
    ?population
    (ROUND(?wikiPersons/?population*1000000) AS ?wikiPersonsPerM)
WHERE
{
    {
        { SELECT ?country (count(*) as ?wikiPersons) WHERE {
            {SELECT DISTINCT ?person ?country WHERE {
                ?country wdt:P31 wd:Q185441 .
                ?person wdt:P31 wd:Q5 .
                ?person wdt:P27 ?country .
                FILTER NOT EXISTS{ ?person wdt:P570 ?date }
            } }
        } GROUP BY ?country }
        ?country wdt:P1082 ?population
        SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
    }
}
ORDER BY DESC(?wikiPersonsPerM)

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object: Not, because correlated variable ?person occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.17 German states, ordered by the number of company headquarters per million inhabitants

```

SELECT
    ?state ?stateLabel
    ?companies
    ?population
    (?companies/?population*1000000 AS ?companiesPerM)
WHERE
{
    {
        { SELECT ?state (count(*) as ?companies) WHERE {
            {SELECT DISTINCT ?company ?state WHERE {
                ?state wdt:P31 wd:Q1221156 .
                ?company wdt:P31/wdt:P279* wd:Q4830453 .
                ?company wdt:P159/wdt:P131* ?state .
                FILTER NOT EXISTS{ ?company wdt:P576 ?date }
            } }
        } GROUP BY ?state }
        ?state wdt:P1082 ?population
        SERVICE wikibase:label { bd:serviceParam wikibase:language "de" }
    }
}
ORDER BY DESC(?companiesPerM)

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.

- B. Differences assuming BNs only as object: Not, because correlated variable ?company occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.18 French communes with names ending in ac

```

SELECT ?item ?itemLabel ?coord
WHERE
{
    ?item    wdt:P31/wdt:P279* wd:Q484170;
            wdt:P17 wd:Q142;
            rdfs:label ?itemLabel;
            wdt:P625 ?coord;
    FILTER (lang(?itemLabel) = "fr").
    FILTER regex (?itemLabel, "ac$").
    FILTER not exists { ?item wdt:P131 wd:Q33788 }
}

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object: Yes, because there is a constant in FILTER NOT EXISTS in object position.
- C. Differences knowing what properties has BNs in Wikidata: Yes, because predicate wdt:P131 has blanks in object position.
- D. Differences in Wikidata: Not, because the following query has result false.

```

ASK {
    ?item    wdt:P31/wdt:P279* wd:Q484170;
            wdt:P17 wd:Q142;
            rdfs:label ?itemLabel;
            wdt:P625 ?coord;
    FILTER (lang(?itemLabel) = "fr").
    FILTER regex (?itemLabel, "ac$").
    { ?item wdt:P131 ?x filter (isBlank(?x)) }
}

```

1.19 Buildings in more than one country

```

SELECT ?item ?itemLabel ?count
WHERE
{
    {
        SELECT ?item (COUNT(DISTINCT ?country) AS ?count) WHERE {
            ?item wdt:P31/wdt:P279* wd:Q41176 .
            ?item wdt:P17 ?country .
        }
    }
}

```

```

    FILTER NOT EXISTS { ?country wdt:P576 ?end }
  }
  GROUP BY ?item
}
FILTER ( ?count > 1 )
SERVICE wikibase:label { bd:serviceParam wikibase:language "en" . }
}
ORDER BY DESC(?count) ?itemL

```

- A. Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object:** Yes, because variable ?country occurs in object position out he FILTER NOT EXISTS.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because predicate wdt:P17 has blanks in object postion.
- D. Differences in Wikidata:** Not, because the following query has result false.

```

ASK {
  ?item wdt:P31/wdt:P279* wd:Q41176 .
  ?item wdt:P17 ?country .
  FILTER (isBlank(?country))
}

```

1.20 List of countries by age of the head of government

```

SELECT DISTINCT ?age ?country ?countryLabel ?hgovernment ?hgovernmentLabel
{
  ?country wdt:P31 wd:Q3624078 .
  FILTER NOT EXISTS {?country wdt:P31 wd:Q3024240}
  ?country p:P6 ?statement .
  ?statement ps:P6 ?hgovernment .
  ?country wdt:P6 ?hgovernment .
  FILTER NOT EXISTS { ?statement pq:P582 ?x }
  ?hgovernment wdt:P569 ?dob . BIND(YEAR(now())-YEAR(?dob) as ?age)
  FILTER(?age>=65)
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}
ORDER BY DESC(?age)

```

- A. Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object:** Not, because variable ?statement occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata:** Not.
- D. Differences in Wikidata:** Not.

1.21 Members of the French National Assembly born out of France

```
SELECT DISTINCT ?item ?itemLabel ?placeLabel ?countryLabel
WHERE
{
  ?item wdt:P39 wd:Q3044918 .
  ?item wdt:P19 ?place .
  ?place wdt:P17 ?country .
  FILTER NOT EXISTS { ?place wdt:P17 wd:Q142 . }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" . }
}
ORDER BY ?countryLabel ?itemLabel
```

- A. Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object:** Yes, because a constant occurs in object position inside the FILTER NOT EXISTS clause.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because predicate wdt:P17 has blanks in object position.
- D. Differences in Wikidata:** Not, because the following query returns false.

```
ASK {
  ?item wdt:P39 wd:Q3044918 .
  ?item wdt:P19 ?place .
  ?place wdt:P17 ?country .
  { ?place wdt:P17 ?x . FILTER (isBlank(?x)) }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" . }
}
```

1.22 Female scientists with most number of sitelinks (but not English Wikipedia)

```
PREFIX schema: <http://schema.org/>

SELECT ?item ?itemLabel ?linkcount WHERE {
  ?item wdt:P31 wd:Q5 .
  ?item wdt:P21 wd:Q6581072 .
  ?item wdt:P106 wd:Q901 .
  ?item wikibase:sitelinks ?linkcount .
  FILTER (?linkcount >= 1) .
  FILTER NOT EXISTS {
    ?article schema:about ?item .
    ?article schema:inLanguage "en" .
    ?article schema:isPartOf <https://en.wikipedia.org/>
  }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en,de,es,ar,fr" }
}
GROUP BY ?item ?itemLabel ?linkcount
ORDER BY DESC(?linkcount)
```

- A. **Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. **Differences assuming BNs only as object:** Yes, because variable `?item` and constants occur in object position inside the FILTER NOT EXISTS clause.
- C. **Differences knowing what properties has BNs in Wikidata:** Not, because the involved predicates have no blanks in object position.
- D. **Differences in Wikidata:** Not.

1.23 List of authors unsuccessfully nominated for Nobel prize in literature

```

SELECT ?nominee ?nomineeLabel
      (SAMPLE(?citizenshipLabel) AS ?country)
      (COUNT(DISTINCT ?year) as ?timesNominated)
WHERE
{
  BIND( wd:Q37922 as ?prize )
  ?nominee p:P1411 [ ps:P1411 ?prize; pq:P585 ?time ]
  BIND( year(?time) as ?year )
  OPTIONAL {
    ?nominee wdt:P27 [ rdfs:label ?citizenshipLabel ]
    filter (lang(?citizenshipLabel) = "en") .
  }
  FILTER NOT EXISTS { ?nominee wdt:P166 ?prize . }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}
GROUP BY ?nominee ?nomineeLabel
ORDER BY DESC(?timesNominated) ?nomineeLabel

```

- A. **Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. **Differences assuming BNs only as object:** Yes because variable `?prize` occurs in object position in the FILTER NOT EXISTS clause and it is correlated with the outer pattern.
- C. **Differences knowing what properties has BNs in Wikidata:** Yes, because predicate `wdt:P166` has blanks in object position.
- D. **Differences in Wikidata:** Not, because the following query returns false.

```

ASK
{
  BIND( wd:Q37922 as ?prize )
  ?nominee p:P1411 [ ps:P1411 ?prize; pq:P585 ?time ]
  BIND( year(?time) as ?year )
  OPTIONAL {
    ?nominee wdt:P27 [ rdfs:label ?citizenshipLabel ]
    filter (lang(?citizenshipLabel) = "en") .
  }
  { ?nominee wdt:P166 ?x . FILTER (isBlank(?x)) }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}

```


1.24 Monuments historiques in Loire-Atlantique

```
SELECT DISTINCT
  ?item
  ?itemLabel
  ?communeLabel
  (group_concat(distinct ?merimee ; separator = ", ") as ?merimee)
  ?coords
  ?image
WHERE
{
  {
    {
      SELECT DISTINCT ?item ?merimee WHERE {
        ?item wdt:P1435/wdt:P279* wd:Q916475 .
        ?item p:P1435 ?heritage_statement .
        FILTER NOT EXISTS { ?heritage_statement pq:P582 ?end . }
        ?item wdt:P380 ?merimee.
      }
      ORDER BY ?merimee
    }
    ?item wdt:P131/wdt:P131* wd:Q3068 .
    ?item wdt:P131 ?commune .
    OPTIONAL { ?item wdt:P625 ?coords . }
    OPTIONAL { ?item wdt:P18 ?image . }
    SERVICE wikibase:label { bd:serviceParam wikibase:language "fr" . }
  }
  GROUP BY ?item ?itemLabel ?communeLabel ?coords ?image
  ORDER BY ?communeLabel ?itemLabel
```

- A. Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object:** Yes because variable ?heritage_statement occurs in object position out the FILTER NOT EXISTS clause.
- C. Differences knowing what properties has BNs in Wikidata:** Not, because predicate p:P1435 has no blanks in object position.
- D. Differences in Wikidata:** Not.

1.25 French people with an article on the English Wikipedia but not on the French one

```
SELECT ?item ?itemLabel (COUNT(distinct ?sitelink) as ?count)
WHERE
{
  ?item wdt:P31 wd:Q5 .
  ?item wdt:P27 wd:Q142 .
  ?sitelink schema:about ?item .
  FILTER EXISTS { ?wen schema:about ?item . ?wen schema:inLanguage "en" }
  FILTER NOT EXISTS { ?wfr schema:about ?item . ?wfr schema:inLanguage "fr" }
  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "fr,en" .
  }
}
```

```

}
GROUP BY ?item ?itemLabel
ORDER BY DESC(?count)

```

A. Differences in any database: Yes, because FILTER NOT EXISTS.

B. Differences assuming BNs only as object: Yes because the following possibilities:

1. Variable ?item in the outer pattern matches a blank.
2. Variable ?item in the inner query matches a blank.
3. Constant "fr" matches a blank.

C. Differences knowing what properties has BNs in Wikidata: Not, because all involved predicates have no blanks in object position.

D. Differences in Wikidata: Not.

1.26 Human gene items with unreferenced statements

```

SELECT ?prop (count(?prop) as ?counts) WHERE
{SELECT DISTINCT ?item ?prop WHERE {
  ?item wdt:P279 wd:Q7187 ;
  wdt:P703 wd:Q5 ;
  ?prop ?statement .
  FILTER(!(regex(str(?prop), "http://www.wikidata.org/prop/direct/" ) ))
  FILTER(regex(str(?prop), "http://www.wikidata.org/prop/" ) )
  FILTER NOT EXISTS {
    ?item ?prop ?statement .
    ?statement prov:wasDerivedFrom ?derivedFrom .}
}
}
GROUP BY ?prop
ORDER BY desc(?counts)

```

A. Differences in any database: Yes, because FILTER NOT EXISTS.

B. Differences assuming BNs only as object: Yes because variable ?statement occurs in object position.

C. Differences knowing what properties has BNs in Wikidata: Yes because property ?prop is undetermined.

D. Differences in Wikidata: Not, because the following query returns false.

```

ASK {
  ?item wdt:P279 wd:Q7187 ;
  wdt:P703 wd:Q5 ;
  ?prop ?statement .
  FILTER(!(regex(str(?prop), "http://www.wikidata.org/prop/direct/" ) ))
  FILTER(regex(str(?prop), "http://www.wikidata.org/prop/" ) )
  { ?item ?prop ?statement . FILTER (isBlank(?statement)) }
}

```

1.27 Human settlements that are not described in the English Wikipedia

```
SELECT DISTINCT ?settlement ?settlementLabel ?coord WHERE {
  VALUES ?type {wd:Q3957 wd:Q515 wd:Q532 wd:Q486972}
  ?settlement wdt:P31 wd:Q3957 ;
              wdt:P625 ?coord .
  ?article schema:about ?settlement .
  FILTER NOT EXISTS { ?wen schema:about ?settlement ; schema:inLanguage "en" }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}
```

A. Differences in any database: Yes, because FILTER NOT EXISTS.

B. Differences assuming BNs only as object: Yes because correlated variable `?settlement` and constant `"en"` occur in object position in the inner pattern.

C. Differences knowing what properties has BNs in Wikidata: Not, because the involved properties have no blank nodes in object position.

D. Differences in Wikidata: Not.

1.28 Human settlements not being described in any language version of Wikipedia

```
SELECT DISTINCT ?city ?cityLabel ?coord WHERE {
  VALUES ?type { wd:Q3957 wd:Q515 wd:Q532 wd:Q486972 } .
  ?city wdt:P31 wd:Q3957 ;
        wdt:P625 ?coord .

  FILTER NOT EXISTS {?article schema:about ?city } .
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" } .
}
```

A. Differences in any database: Yes, because FILTER NOT EXISTS.

B. Differences assuming BNs only as object: Yes because correlated variable `?city` occurs in object position in the inner pattern.

C. Differences knowing what properties has BNs in Wikidata: Not, because property `schema:about` has no blank nodes in object position.

D. Differences in Wikidata: Not.

1.29 A batch of entities that were imported from Commons that would need to be merged with pre-existing entities

```
SELECT ?subject ?subjectLabel ?gallery
WHERE {
  ?subject wdt:P31 wd:Q5 .
  ?subject wdt:P935 ?gallery.
  FILTER NOT EXISTS { ?subject wdt:P18 ?x }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr,en" }
```

```

    FILTER regex(str(?subject), "Q211", "i")
}
LIMIT 100

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object: Not because correlated variable ?subject occurs in subject positions.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.30 How many people are there whose "number of children (P1971)" matches with "child (P40)"?

```

SELECT (COUNT(DISTINCT(?person)) AS ?countPerson)
WHERE {
  {
    ?person wdt:P1971 ?numOfChildren .
    {SELECT ?person (COUNT(?child) AS ?numOfChildrenComputed)
    WHERE {
      ?person wdt:P40 ?child
    }
    GROUP BY ?person}
    FILTER (?numOfChildren = ?numOfChildrenComputed)
  }
  UNION
  {
    ?person wdt:P1971 "0"^^xsd:decimal .
    FILTER NOT EXISTS { ?person wdt:P40 ?child }
  }
}

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object: Not because correlated variable ?person occurs in subject positions.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.31 Instances of actual languages

```

SELECT ?item ?itemLabel
WHERE {
  ?item wdt:P31/wdt:P31* wd:Q34770 .
  FILTER NOT EXISTS {?item wdt:P31 wd:Q34770}
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}

```

- A. **Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. **Differences assuming BNs only as object:** Yes, because a constant occurs in object position inside the FILTER NOT EXISTS clause.
- C. **Differences knowing what properties has BNs in Wikidata:** Yes, because predicate `wdt:P31` has blanks in object position.
- D. **Differences in Wikidata:** Not, because the following query returns false.

```
ASK{
  ?item wdt:P31/wdt:P31* wd:Q34770 .
  {?item wdt:P31 ?x FILTER (isBlank(?x))}
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}
```

1.32 People that have a label in Hebrew, but their given name doesn't have a label in Hebrew

```
SELECT ?item ?itemLabelHE ?name WHERE {
  ?item wdt:P31 wd:Q5.
  ?item wdt:P735 ?name.
  FILTER NOT EXISTS {?name rdfs:label ?nameLabelHE filter (lang(?nameLabelHE) = "he")} .
  ?item rdfs:label ?itemLabelHE filter (lang(?itemLabelHE) = "he") .
}
```

- A. **Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. **Differences assuming BNs only as object:** Yes, because a blank can match variable `?name` in the outer pattern or a blank can match variable `?nameLabelHE` in the inner pattern.
- C. **Differences knowing what properties has BNs in Wikidata:** Yes, because predicate `wdt:P735` has blanks in object position.
- D. **Differences in Wikidata:** Yes, because the following query returns true.

```
ASK {
  ?item wdt:P31 wd:Q5.
  ?item wdt:P735 ?name.
  ?item rdfs:label ?itemLabelHE filter (lang(?itemLabelHE) = "he") .
  FILTER (isBlank(?name))
}
```

1.33 Most specific parent classes of Wikidata classes

```

SELECT ?lcs ?lcsLabel WHERE {
  ?lcs ^wdt:P279* wd:Q32815, wd:Q34627, wd:Q16970, wd:Q16560 .
  filter not exists {
    ?sublcs ^wdt:P279* wd:Q32815, wd:Q34627, wd:Q16970, wd:Q16560 ;
    wdt:P279 ?lcs . }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE], en" . }
}

```

- A. Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object:** Yes, because variable ?lcs and several constants occur in object position in the FILTER NOT EXISTS clause.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because predicate wdt:P279 has blanks in object position.
- D. Differences in Wikidata:** Not, because the following query returns false.

```

ASK {
  ?lcs ^wdt:P279* wd:Q32815, wd:Q34627, wd:Q16970, wd:Q16560 .
  ?sublcs ^wdt:P279* wd:Q32815, wd:Q34627, wd:Q16970, wd:Q16560 ;
  wdt:P279 ?x .
  FILTER (isBlank(?x))
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE], en" . }
}

```

1.34 List of present-day countries and capital(s)

```

SELECT DISTINCT ?country ?countryLabel ?capital ?capitalLabel
WHERE
{
  ?country wdt:P31 wd:Q3624078 .
  FILTER NOT EXISTS {?country wdt:P31 wd:Q3024240}
  FILTER NOT EXISTS {?country wdt:P31 wd:Q28171280}
  OPTIONAL { ?country wdt:P36 ?capital } .
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}
ORDER BY ?countryLabel

```

- A. Differences in any database:** Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object:** Yes, because constants occur in object position in the FILTER NOT EXISTS clause.
- C. Differences knowing what properties has BNs in Wikidata:** Yes, because predicate wdt:P31 has blanks in object position.
- D. Differences in Wikidata:** Not, because the following query returns false.

```

ASK
{
  ?country wdt:P31 wd:Q3624078 .
  ?country wdt:P31 ?x .
  FILTER (isBlank(?x))
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}

```

1.35 Mushers with neither a ranking in a race nor a reason for not finishing it

```

SELECT distinct ?race ?raceLabel ?musherLabel
WHERE
{
  { ?race wdt:P31/wdt:P279* wd:Q1968664 . }
  UNION { ?race wdt:P31/wdt:P641* wd:Q1968664 . }
  ?race p:P710 ?musherS .
  ?musherS ps:P710 ?musher .
  FILTER NOT EXISTS { ?musherS pq:P1352 ?rank }
  FILTER NOT EXISTS { ?musherS pq:P793 ?event }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "fr,en" }
}
ORDER BY ?raceLabel

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS.
- B. Differences assuming BNs only as object: Not, because variable ?musherS occurs in subject position in the outer and in the inner pattern.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.36 Occupations with a German label

```

SELECT DISTINCT ?occ ?occLabelEN ?occLabelDE {
  ?occ (wdt:P31/wdt:P31* | wdt:P279/wdt:P279*) wd:Q12737077 .
  FILTER NOT EXISTS { ?occ wdt:P31 wd:Q5 }
  FILTER NOT EXISTS { ?occ wdt:P31 wd:Q13406463 }
  FILTER ( ?occ != wd:Q28640 )
  ?occ rdfs:label ?occLabelDE filter (lang(?occLabelDE) = "de") .
  OPTIONAL {
    ?occ rdfs:label ?occLabelEN filter (lang(?occLabelEN) = "en")
  }
}

```

- A. Differences in any database: Yes, because FILTER NOT EXISTS and !=.
- B. Differences assuming BNs only as object: Yes, because constants occur in object position inside the FILTER NOT EXISTS clause. Regarding to !=, it does not matters because the variable involved occurs in a subject position in the outer pattern.

C. Differences knowing what properties has BNs in Wikidata: Yes, because predicate `wdt:P31` has blanks in object position.

D. Differences in Wikidata: Not, because the following query returns false.

```
ASK {
  ?occ (wdt:P31/wdt:P31* | wdt:P279/wdt:P279*) wd:Q12737077 .
  ?occ wdt:P31 ?x .
  ?occ rdfs:label ?occLabelDE filter (lang(?occLabelDE) = "de") .
  FILTER ( ?occ != wd:Q28640 )
  FILTER (isBlank(?x))
}
```

1.37 Women born in Wales without an article in Welsh for Listeria

```
SELECT ?item
WHERE
{
  ?item wdt:P21 wd:Q6581072 .

  { ?item wdt:P19 wd:Q25 }
  UNION { ?item wdt:P19 ?pob . ?pob wdt:P131* wd:Q25 }

  OPTIONAL { ?sitelink schema:about ?item . ?sitelink schema:inLanguage "cy" }
  FILTER (!BOUND(?sitelink))

  ?item wdt:P31 wd:Q5 .
}
LIMIT 1000
```

A. Differences in any database: Yes, because there is a negation based on `OPTIONAL+!BOUND`.

B. Differences assuming BNs only as object: Yes, because variable `?item` and constant `"cy"` occur in object position inside the `OPTIONAL` clause.

C. Differences knowing what properties has BNs in Wikidata: Not, because predicates `schema:about` and `schema:inLanguage` do not have blanks in object position.

D. Differences in Wikidata: Not.

1.38 People born before year 1880 with no death date

```
SELECT ?h ?date
WHERE
{
  ?h wdt:P31 wd:Q5 .
  ?h wdt:P569 ?date .
  OPTIONAL {?h wdt:P570 ?d }
  FILTER (?date < "1880-01-01T00:00:00Z"^^xsd:dateTime)
  FILTER (!bound(?d))
}
```



```
}  
LIMIT 100
```

- A. Differences in any database: Yes, because there is a negation based on OPTIONAL+!BOUND.
- B. Differences assuming BNs only as object: Not, because correlated variable occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.39 Actresses without a description in Spanish

```
SELECT DISTINCT ?item ?itemLabel  
WHERE  
{  
  ?item wdt:P106 wd:Q33999 .  
  ?item wdt:P21 wd:Q6581072 .  
  OPTIONAL { ?item schema:description ?itemDescription }  
  FILTER (!BOUND(?itemDescription))  
  ?item wdt:P31 wd:Q5 .  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "es"}  
}  
LIMIT 111
```

- A. Differences in any database: Yes, because there is a negation based on OPTIONAL+!BOUND.
- B. Differences assuming BNs only as object: Not, because correlated variable occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.40 Chemical compounds with no CAS registry number

```
SELECT DISTINCT ?compound ?compoundLabel ?formula ?inchikey ?chemspider ?pubchem WHERE {  
  ?compound wdt:P31 wd:Q11173 .  
  OPTIONAL { ?compound wdt:P231 ?cas . }  
  OPTIONAL { ?compound wdt:P274 ?formula . }  
  OPTIONAL { ?compound wdt:P235 ?inchikey . }  
  OPTIONAL { ?compound wdt:P661 ?chemspider . }  
  OPTIONAL { ?compound wdt:P662 ?pubchem . }  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }  
  FILTER(!BOUND(?cas))  
}
```

- A. Differences in any database: Yes, because there is a negation based on OPTIONAL+!BOUND.
- B. Differences assuming BNs only as object: Not, because correlated variable occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.41 Most frequent occupations without a label in a given language

```
SELECT ?occup ?occupLabel ?count
WHERE
{
  {
    SELECT ?occup (COUNT(?person) as ?count)
    WHERE
    {
      ?person wdt:P106 ?occup
    }
    GROUP BY ?occup
    ORDER BY DESC(?count)
    LIMIT 1000
  }
  OPTIONAL {?occup rdfs:label ?label1 filter(lang(?label1) = "lv")}
  FILTER(!BOUND(?label1))

  SERVICE wikibase:label { bd:serviceParam wikibase:language "en,es,pl,ja,zh,ru" }
}
ORDER BY DESC(?count)
LIMIT 50
```

- A. Differences in any database: Yes, because there is a negation based on OPTIONAL+!BOUND.
- B. Differences assuming BNs only as object: Not, because correlated variable occurs in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.42 Yearly Population stacked by country

```
SELECT ?year (AVG(?pop) AS ?population) ?countryLabel
(COUNT(*) AS ?number_of_chosen_sources) (SAMPLE(?method) AS ?a_source_of_those_chosen)
WHERE
{
  ?country wdt:P31 wd:Q6256;
    p:P1082 ?popStatement .
  ?popStatement ps:P1082 ?pop;
    pq:P585 ?date .
  BIND(STR(YEAR(?date)) AS ?year)

  OPTIONAL { ?popStatement pq:P459 ?method. }
  OPTIONAL { ?country p:P1082 [ pq:P585 ?d; pq:P459 ?estimate ].
    FILTER(STR(YEAR(?d)) = ?year). FILTER(?estimate = wd:Q791801). }
  OPTIONAL { ?country p:P1082 [ pq:P585 ?e; pq:P459 ?census ].
    FILTER(STR(YEAR(?e)) = ?year). FILTER(?census = wd:Q39825). }
  OPTIONAL { ?country p:P1082 [ pq:P585 ?f; pq:P459 ?other ].
    FILTER(STR(YEAR(?f)) = ?year). FILTER(?other != wd:Q39825 && ?other != wd:Q791801). }
  BIND(COALESCE(
    IF(BOUND(?census), ?census, 1/0),
    IF(BOUND(?other), ?other, 1/0),
```

```

    IF(BOUND(?estimate), ?estimate, 1/0) ) AS ?pref_method).
    FILTER(IF(BOUND(?pref_method),?method = ?pref_method,true))

    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en" }
}
GROUP BY ?year ?countryLabel
ORDER BY ?year ?countryLabel

```

- A. Differences in any database: Yes, because !=.
- B. Differences assuming BNs only as object: Yes, because variable ?other occurs in object position.
- C. Differences knowing what properties has BNs in Wikidata: Yes, because predicate pq:P459 has blanks in object position.
- D. Differences in Wikidata: Not, because the following query returns false.

```

ASK
{
  ?country wdt:P31 wd:Q6256; p:P1082 ?popStatement .
  ?popStatement ps:P1082 ?pop; pq:P585 ?date .
  { ?country p:P1082 [ pq:P459 ?other ]. FILTER(isBlank(?other)). }
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en" }
}

```

1.43 Candidates for the Dutch general election 2017 living abroad

```

SELECT ?item ?itemLabel ?coordinates WHERE {
  ?item p:P3602 ?node.
  ?item wdt:P551 ?residence .
  ?residence wdt:P17 ?country ;
             wdt:P625 ?coordinates .
  ?node ps:P3602 wd:Q16061881.
  FILTER (?country != wd:Q55)
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],nl". }
}

```

- A. Differences in any database: Yes, because !=.
- B. Differences assuming BNs only as object: Yes, because variable ?country occurs in object position.
- C. Differences knowing what properties has BNs in Wikidata: Yes, because predicate pq:P17 has blanks in object position.
- D. Differences in Wikidata: Not, because the following query returns false.

```

ASK {
  ?item p:P3602 ?node.
  ?item wdt:P551 ?residence .
  ?residence wdt:P17 ?country ;
             wdt:P625 ?coordinates .
  ?node ps:P3602 wd:Q16061881.
  FILTER (isBlank(?country))
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],nl". }
}

```

1.44 Metabolite-metabolite interactions (mostly conversions) and their pKa change

```

PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wp:   <http://vocabularies.wikipathways.org/wp#>
PREFIX dcterms: <http://purl.org/dc/terms/>
SELECT DISTINCT ?wpid ?metaboliteLabel ?pKa ?pKa2 ?metabolite2Label ?wpid2
WITH {
  SELECT ?wpid ?source_pathway ?metabolite ?pKa ?pKa2 ?metabolite2 ?wpid2
  WHERE {
    ?pathway wdt:P2410 ?wpid ;
             wdt:P527 ?metabolite ;
             wdt:P2888 ?source_pathway .
    ?metabolite wdt:P1117 ?pKa .

    SERVICE <http://sparql.wikipathways.org/> {
      ?wp_mb1 wp:bdbWikidata ?metabolite .
      ?wp_mb1 dcterms:isPartOf ?interaction .
      ?interaction rdf:type wp:Interaction .
      ?wp_mb2 dcterms:isPartOf ?interaction .
      ?wp_mb2 wp:bdbWikidata ?metabolite2 .
      FILTER (?wp_mb1 != ?wp_mb2)
    }
  }
} AS %result
WHERE {
  INCLUDE %result
  ?metabolite2 wdt:P1117 ?pKa2 .
  ?pathway2 wdt:P2410 ?wpid2 ;
            wdt:P527 ?metabolite2 ;
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
}

```

- A. Differences in any database: Yes, because !=.
- B. Differences assuming BNs only as object: Not, because variables ?wp_mb1 and ?wp_mb2 occur in subject position.
- C. Differences knowing what properties has BNs in Wikidata: Not.
- D. Differences in Wikidata: Not.

1.45 Chemical compounds in Wikidata sharing the same CAS registry number

```
SELECT DISTINCT ?cas ?compound1 ?compound1Label ?compound2 ?compound2Label WHERE {
  ?compound1 wdt:P231 ?cas .
  ?compound2 wdt:P231 ?cas .
  FILTER (?compound1 != ?compound2)
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
}
```

A. Differences in any database: Yes, because !=.

B. Differences assuming BNs only as object: Not, because variables ?compound1 and ?compound2 occur in subject position.

C. Differences knowing what properties has BNs in Wikidata: Not.

D. Differences in Wikidata: Not.

1.46 Objects that have globe which does not match their coordinates

```
SELECT ?item ?body ?globe
WHERE {
  ?item wdt:P376 ?body .
  ?item p:P625/psv:P625/wikibase:geoGlobe ?globe .
  filter (?globe != ?body) .
} LIMIT 30
```

A. Differences in any database: Yes, because !=.

B. Differences assuming BNs only as object: Yes, because involved variables occur in object position.

C. Differences knowing what properties has BNs in Wikidata: Not, because predicates wdt:P376 and wikibase:geoGlobe have no blanks in object position.

D. Differences in Wikidata: Not.

1.47 Mountains, with coordinates, not located on Earth

```
SELECT DISTINCT ?item ?name ?coord ?lat ?lon ?globe
{
  ?item wdt:P31 wd:Q8502 ;
    p:P625 [
      psv:P625 [
        wikibase:geoLatitude ?lat ;
        wikibase:geoLongitude ?lon ;
        wikibase:geoGlobe ?globe ;
      ] ;
      ps:P625 ?coord
    ]
  FILTER ( ?globe != wd:Q2 )
  SERVICE wikibase:label {
```

```
    bd:serviceParam wikibase:language "en" .
    ?item rdfs:label ?name
  }
}
ORDER BY ASC (?name)
```

- A. **Differences in any database:** Yes, because !=.
- B. **Differences assuming BNs only as object:** Yes, because involved variables occur in object position.
- C. **Differences knowing what properties has BNs in Wikidata:** Not, because predicate `wikibase:geoGlobe` has no blanks in object position.
- D. **Differences in Wikidata:** Not.

2 Summary

Query	A	B	C	D	Features
Women with most sitelinks and no image born in 1921 or later	Y	N	N	N	MINUS
Contemporary Indian actresses	Y	N	N	N	MINUS
Map of places mentioned in travel stories with text in French...	Y	Y	Y	N	MINUS
Streets in France without a city	Y	N	N	N	MINUS
ATMs around Munich belonging to the Bankcard-Servicenet...	Y	N	N	N	MINUS
Objects with most mass	Y	Y	Y	N	MINUS
Periods of Japanese history and what they were named after	Y	Y	Y	N	MINUS
Actors who played the same role more than 40 years apart	Y	Y	Y	N	MINUS, !=
Sandwich ingredients	Y	Y	Y	N	MINUS
Odd countries	Y	Y	Y	Y	MINUS
Capitals that aren't capitals	Y	Y	Y	Y	MINUS
Look up q-codes based on English description	Y	Y	Y	N	MINUS
Metabolites and the species where they are found in	Y	Y	Y	N	MINUS
Largest cities with female mayor	Y	N	N	N	FILTER NOT EXISTS
List of countries ordered by the number of their cities with...	Y	N	N	N	FILTER NOT EXISTS
Wikidata people per million inhabitants for all EU countries	Y	N	N	N	FILTER NOT EXISTS
German states, ordered by the number of company headquar...	Y	N	N	N	FILTER NOT EXISTS
French communes with names ending in ac	Y	Y	Y	N	FILTER NOT EXISTS
Buildings in more than one country	Y	Y	Y	N	FILTER NOT EXISTS
List of countries by age of the head of government	Y	N	N	N	FILTER NOT EXISTS
Members of the French National Assembly born out of France	Y	Y	Y	N	FILTER NOT EXISTS
Female scientists with most number of sitelinks	Y	Y	N	N	FILTER NOT EXISTS
List of authors unsuccessfully nominated for Nobel prize in...	Y	Y	Y	N	FILTER NOT EXISTS
Monuments historiques in Loire-Atlantique	Y	Y	N	N	FILTER NOT EXISTS
French people with an article on the English Wikipedia but...	Y	Y	N	N	FILTER NOT EXISTS
Human gene items with unreferenced statements	Y	Y	Y	N	FILTER NOT EXISTS
Human settlements that are not described in the English...	Y	Y	N	N	FILTER NOT EXISTS
Human settlements not being described in any language...	Y	Y	N	N	FILTER NOT EXISTS
A batch of entities that were imported from Commons that...	Y	N	N	N	FILTER NOT EXISTS
How many people are there whose "number of children...	Y	N	N	N	FILTER NOT EXISTS
Instances of actual languages	Y	Y	Y	N	FILTER NOT EXISTS
People that have a label in Hebrew, but their given name...	Y	Y	Y	Y	FILTER NOT EXISTS
Most specific parent classes of Wikidata classes	Y	Y	Y	N	FILTER NOT EXISTS
List of present-day countries and capital(s)	Y	Y	Y	N	FILTER NOT EXISTS
Mushers with neither a ranking in a race nor a reason for...	Y	N	N	N	FILTER NOT EXISTS
Occupations with a German label	Y	Y	Y	N	FILTER NOT EXISTS
Women born in Wales without an article in Welsh for Listeria	Y	Y	N	N	OPTIONAL+!BOUND
People born before year 1880 with no death date	Y	N	N	N	OPTIONAL+!BOUND
Actresses without a description in Spanish	Y	N	N	N	OPTIONAL+!BOUND
Chemical compounds with no CAS registry number	Y	N	N	N	OPTIONAL+!BOUND
Most frequent occupations without a label in a given language	Y	N	N	N	OPTIONAL+!BOUND
Yearly Population stacked by country	Y	Y	Y	N	!=
Candidates for the Dutch general election 2017 living abroad	Y	Y	Y	N	!=
Metabolite-metabolite interactions (mostly conversions) and...	Y	N	N	N	!=
Chemical compounds in Wikidata sharing the same CAS...	Y	N	N	N	!=
Objects that have globe which does not match their coordinates	Y	Y	N	N	!=
Mountains, with coordinates, not located on Earth	Y	Y	N	N	!=

Feature	A	B	C	D
MINUS	13	9	9	2
FILTER NOT EXISTS	23	15	10	1
OPTIONAL+!BOUND	5	1	0	0
!=	7	5	3	0
Total	47	29	21	3

(From 446 example queries)